

金莉丽 陈树康 编著

BASIC

物理化学 微型机计算程序



华中工学院出版社

物理化学微型机計算程序

金丽莉 陈树康 编著

华中工学院出版社

内 容 提 要

本书从实际应用出发,结合常用的数值计算方法,介绍了用微型计算机进行物理化学实验数据的处理和理论计算的方法,并提供了用BASIC语言编写的有关化学热力学、动力学、晶体化学和量子化学方面的一些典型程序。针对化学工作者对计算数学不太熟悉的情况,本书力求通俗易懂,深入浅出,与实例紧密结合,以适合广大化学工作者的需要。

本书可作为物理化学实验教材的一部分,供物化专业师生使用,也可供化学类其它专业的科技工作者和师生参考。

物理化学微型机计算程序

金丽莉 陈树康 编著

责任编辑 汪玉山

华中工学院出版社出版发行

(武昌喻家山)

新华书店湖北发行所经销

华中工学院出版社沔阳印刷厂印刷

开本: 787×1092 1/32 印张: 6.25 字数: 138,000

1987年元月第1版 1987年元月第1次印刷

印数: 1—2,000

ISBN 7—5609—0090—9/O·12

统一书号: 13255-060 定价: 1.10 元

前 言

微型计算机的迅速普及,使每个化学工作者都面临着如何尽快地把微型机应用于化学教学和科研领域的问题。对物理化学这门学科来说,实验数据的处理和理论计算是该学科教学和科研的重要环节,因此,将微型机应用于物理化学显得尤为迫切。本书就是为适应这种需要而编写的。

化学工作者使用计算机的一个主要障碍是对计算数学不太熟悉。一般有关计算数学的书籍往往侧重于严格的理论证明和数学推导,使化学工作者不易接受。本书试图在两者之间架起一座桥梁。把物理化学中最常用和最基本的算法分类叙述,结合实例,编写了相应的 BASIC 程序。在理论叙述上力求通俗易懂,深入浅出,使其容易为广大化学工作者和学生所接受。对于那些多次用到的算法则编出通用子程序。书中所举实例多为物理化学实验数据的处理和结构化学计算,内容涉及化学热力学、化学反应动力学、晶体化学和量子化学等。

本书是编者近几年来在物理化学和微型机方面从事教学的经验总结。书中全部程序均在 APPLE-Ⅰ 和 TRS-80 机上实算通过,对于其它机型用户可以适当调整程序,使其更为简便适用。并请用户注意书中的有关注释。

本书一至六章由金丽莉编写。七、八两章及一至三章部分小节由陈树康编写。第八章中量子化学程序由中国科学院上海药物研究所吴吉安同志提供。周卓媛同志参加了部分程序的调试工作。书稿由任建国老师审定,查全性教授为本书的修改提

出了宝贵意见，龚本玲副教授对于本书的编写始终给予热情的支持，在此，我们谨向各位师长致以衷心的感谢。

由于我们实践经验不足，水平有限，书中不当之处，欢迎批评指正。

编者

一九八五年三月

目 录

第一章	由确定公式计算实验结果	(1)
§ 1.1	电导法测定弱电解质电离常数	(1)
§ 1.2	离子迁移数的测定——希托夫法	(4)
§ 1.3	凝固点降低法测分子量	(8)
§ 1.4	磁化率的测定	(11)
§ 1.5	X射线粉末图的分析	(15)
第二章	直线拟合与化学热力学参数的确定	(21)
§ 2.1	基本算法	(21)
§ 2.2	最小二乘直线拟合子程序及说明	(25)
§ 2.3	一级反应——蔗糖转化	(28)
§ 2.4	粘度法测高聚物分子量	(30)
§ 2.5	液体饱和和蒸气压的测定	(33)
§ 2.6	偶极矩的测定	(36)
§ 2.7	HCl红外光谱分析	(41)
第三章	线性模型中物理化学参数的确定	(46)
§ 3.1	线性联立方程组的数值解法	(46)
§ 3.2	线性模型中参数的确定	(50)
§ 3.3	丙酮溴化	(53)
§ 3.4	线性模型的推广——多项式拟合	(59)
§ 3.5	氢原子光谱和钠原子光谱的分析	(61)
第四章	非线性模型与物理化学参数的确定	(66)
§ 4.1	高斯-牛顿法基本原理	(66)
§ 4.2	溶液表面张力的测定——气泡最大压力法	(69)
§ 4.3	BET方程的非线性拟合	(74)
第五章	数值积分及其应用	(80)
§ 5.1	梯形法则	(80)

§ 5.2	辛普森法则	(82)
§ 5.3	自动加密法	(83)
§ 5.4	量热计体系温度改变值的确定	(85)
第六章	常微分方程的数值解法与化学反应动力学过程模拟	(93)
§ 6.1	初值问题与化学反应动力学方程	(94)
§ 6.2	边值问题与化学反应动力学方程	(105)
第七章	晶体化学计算	(116)
§ 7.1	倒易点阵和晶胞体积	(116)
§ 7.2	晶面指标和晶面间距	(119)
§ 7.3	衍射指标和Bragg角	(120)
§ 7.4	原子间距离、键长和键角	(122)
§ 7.5	晶体坐标转换为直角坐标	(124)
§ 7.6	晶体化学计算程序及使用说明	(126)
第八章	量子化学计算	(134)
§ 8.1	原子坐标	(134)
§ 8.2	重叠积分	(140)
§ 8.3	库仑积分	(145)
§ 8.4	实对称矩阵的特征值和特征向量	(148)
§ 8.5	计算结果输出	(152)
§ 8.6	简单Hückel分子轨道法(HMO)	(154)
§ 8.7	推广Hückel分子轨道法(EHMO)	(172)
§ 8.8	全略微分重叠法(CNDO/2)	(181)
主要参考资料		(194)

第一章 由确定公式计算实验结果

在物理化学实验中，有一部分是属于测量量与待求量之间已有确定的函数关系，只要把测得的数据代入已知公式中，经计算就可以得出结果。由于这种计算往往要重复多次，使用计算机处理则是十分方便的。我们只需按公式编出程序，送入实验数据即可得到结果。

§ 1.1 电导法测定弱电解质电离常数

以HAc电离常数的测定为例，使用DDS-11A型电导仪，可直接测得不同浓度HAc溶液的比电导 L_0 ，由公式：

$$\Lambda = L_0 \cdot 1000 / C \quad (1)$$

计算浓度为C时溶液的当量电导 Λ 。已知298K时无限稀释HAc溶液的当量电导 $\Lambda_0 = 3.908 \times 10^{-2} (\text{S} \cdot \text{m}^2 \cdot \text{mol}^{-1})$ ，则电离度 α 及电离常数 K 。分别由下面两式计算：

$$\alpha = \Lambda / \Lambda_0 \quad (2)$$

$$K_0 = \frac{C \Lambda^2}{\Lambda_0 (\Lambda_0 - \Lambda)} \quad (3)$$

实验记录的初始数据有：HAc的浓度C，与其相应的比电导值 L_0 ，一共记录了N组C- L_0 数据，由此求算电离常数 K_0 。下面介绍计算程序。

程序名：CONDUCT

程序中各变量的意义

简单变量:

N——实验数据点数。

KC——电离常数的平均值。

下标变量:

B(0)——存放无限稀释溶液的当量电导 Λ_0 ($S \cdot m^2 \cdot mol^{-1}$)。

一维数组:

C(N)——存放浓度数据 (mol/L)。

LO(N)——存放比电导数据 ($S \cdot m^{-1}$)。

B(N)——存放当量电导 Λ ($S \cdot m^2 \cdot mol^{-1}$)。

K(N)——存放每次计算出的电离常数值。

R(N)——存放电离度 α 。

程序清单

```
10 INPUT "SET OF DATA : N = ", N
20 DIM C(N), LO(N), K(N), R(N), B(N)
30 K = 0
35 INPUT "EQUIVALENT CONDUCTIVITY AT
INFINITE DILUTION B(0) = ", B(0)
40 PRINT "C", "LO", "B/B(0)", "K"
50 FOR I = 1 TO N
60 READ C(I), LO(I)
70 B(I) = LO(I) / C(I) * 1E-3
80 R(I) = B(I) / B(0)
90 K(I) = C(I) * B(I) * B(I) / (B(0) * (B(0) - B(I)))
100 K = K + K(I)
110 PRINT C(I), LO(I), R(I), K(I)
120 NEXT I
130 KC = K / N : PRINT "KC = ", KC
```

150 DATA (注1)

160 END

还可以用下述方法处理该实验的数据：将电离常数计算公式(3)改写成如下形式：

$$C\Lambda = \Lambda_0^2 K_0 \cdot \frac{1}{\Lambda} - K_0 \Lambda_0 \quad (4)$$

$C\Lambda$ 与 $1/\Lambda$ 为直线关系，利用第二章介绍的算法做直线拟合，由斜率 $\Lambda_0^2 K_0$ 计算 K_0 值。读者参阅第二章后，可自编这段程序。

例 在298K温度下测HAc溶液电离常数，记录了如下数据：

浓度C	0.0200	0.0100	0.0050
(mol/L)			
比电导 L_0	2.345×10^{-2}	1.530×10^{-2}	1.020×10^{-2}
($S \cdot m^{-1}$)			
浓度C	0.0025	0.00125	
比电导 L_0	0.680×10^{-2}	0.434×10^{-2}	

利用程序CONDUC，将以上数据按 $C_1, L_{01}, C_2, L_{02}, \dots$ 的顺序写入150句内，由键盘送入N与B(0)的值。本例中 $N=5, B(0)=3.908 \times 10^{-2} (S \cdot mol^{-1} \cdot m^2)$

运行结果(注2)

C	LO	B/B(0)	K
.02	.02345	.0300025589	1.85599157E-05

(注1) 程序中的DATA语句，供用户写入自己的实验数据(本例中为 $C-L_0$)，请用户注意数据排列顺序必须与READ语句相对应。以后各章中的程序均作同样处理。

(注2) 程序运行结果均为计算机打印的数值，没有按实验本身的有效数字取舍，以后各例均如此，请读者注意。

.01	.0153	.0391504606	1.59521185E-05
5E-03	.0102	.0522006142	1.43748992E-05
2.5E-03	6.8E-03	.0696008189	1.30166548E-05
1.25E-03	4.34E-03	.0888433982	1.08284753E-05
KC = 1.45464127E-05			

§ 1.2 离子迁移数的测定——希托夫法

本例以稀 H_2SO_4 溶液 (约 0.02mol/L) 为待测体系, 用气体库仑计测定总电量。根据定义, 某离子的迁移数就是该离子输送的电量与通过溶液的总电量之比。离子输送电量的法拉第数等于同名电极区浓度减少 (或增加) 的当量数; 总电量的法拉第数等于气体库仑计中生成的 H_2 (或 O_2) 的当量数。阴离子的迁移数 t_- 可以由下式计算:

$$t_- = \frac{\text{阴极区 } \text{H}_2\text{SO}_4 \text{ 减少的当量数}}{\text{库仑计中生成的 } \text{H}_2 \text{ (或 } \text{O}_2 \text{) 的当量数}} \quad (5)$$

阳离子的迁移数 t_+ 可以有两种算法:

第一种是由 $t_+ = 1 - t_-$ 来计算;

第二种是由阳极区测得的数据进行计算:

$$t_+ = \frac{\text{阳极区 } \text{H}_2\text{SO}_4 \text{ 增加的当量数}}{\text{库仑计中生成的 } \text{H}_2 \text{ (或 } \text{O}_2 \text{) 的当量数}} \quad (6)$$

初始数据以及计算过程如下:

温度 T (K), 大气压 P (mmHg), 水的饱和蒸气压 P_w (mmHg), 气体库仑计读数 $V_{\text{始}}$ 、 $V_{\text{终}}$ 。

混合气体体积 $V = V_{\text{终}} - V_{\text{始}}$ 。

根据法拉第定律, 混合气体的电化学当量为:

$$Q = \frac{4}{3} \cdot \frac{(P - P_w)}{RT} V \quad (7)$$

Q就是总电量（法拉第数）。

通电前 H_2SO_4 溶液的有关数据有：

10ml H_2SO_4 溶液重 W_0 g；滴定消耗标准 NaOH 溶液 V_{NaOH} (ml)，浓度为 N_{NaOH} (mol)。由此算出通电前10ml H_2SO_4 溶液中含 H_2SO_4 的当量数为：

$$n_0 = \frac{N_{\text{NaOH}} \cdot V_{\text{NaOH}}}{V_{\text{H}_2\text{SO}_4}} \cdot \frac{10}{1000} \quad (8)$$

其中含水：

$$W_{0\text{水}} = W_0 - n_0 \cdot M_{\text{H}_2\text{SO}_4} / 2$$

每克水含 H_2SO_4 的当量数为： $n_0 / W_{0\text{水}}$ 。

通电后阴极区的有关数据：

阴极区溶液总重量 $W_{\text{总}}$ (g)；10ml溶液重 W_1 (g)；滴定消耗标准 NaOH 体积 V (ml)。

由下式计算在通电后10ml溶液中含 H_2SO_4 的当量数 n_1 ：

$$n_1 = \frac{N_{\text{NaOH}} \cdot V_{\text{NaOH}}}{V_{\text{H}_2\text{SO}_4}} \cdot \frac{10}{1000} \quad (9)$$

阴极区含水量： $W_{1\text{水}} = W_1 - n_1 \cdot M_{\text{H}_2\text{SO}_4} / 2$ ；每克水含 H_2SO_4 的当量数为： $n_1 / W_{1\text{水}}$ ；阴极区总水量： $W_{\text{阴水}} = W_{\text{总}} \cdot W_{1\text{水}} / W_1$ 。然后根据(5)式计算 SO_4^{2-} 的迁移数：

$$t_{\text{SO}_4^{2-}} = \frac{n_0 / W_{0\text{水}} - n_1 / W_{1\text{水}}}{Q}$$

H^+ 的迁移数为： $t_{\text{H}^+} = 1 - t_{\text{SO}_4^{2-}}$

程序名：HITTORF

程序中各变量的意义：

T——室温(℃)。

P——大气压(mmHg)。

V——混合气体体积(ml)。

$R = 62400(\text{mmHg} \cdot \text{ml} \cdot \text{K}^{-1} \cdot \text{mol}^{-1})$ 。

Q——总电量（法拉第数）。

G₀、G₁——分别表示通电前后每克水中含H₂SO₄的当量数。

N₀、N₁——分别表示通电前后10ml溶液中含H₂SO₄的当量数（参见公式(8)、(9)）。

M——H₂SO₄的分子量（M=98）。

WWH——通电前10ml溶液含水量。

WA——阴极区溶液总含水量。

WH——10ml阴极区溶液中的含水量。

TRAN——SO₄⁻的迁移数。

其它各量如 NNA、WO、VO、WS、W₁、V₁等，程序中已有说明。

程序清单

```
10 INPUT "T=" ; T : INPUT "P=" ; P
20 INPUT "VAPOR PRESSURE OF WATER=" ; PW
30 INPUT "V=" ; V
40 R = 62400 : Q = 4 * (P - PW) * V / (3 * R * (T + 273))
60 PRINT "FARADAYS OF Q=" Q
70 INPUT "CONCENTRATION OF NAOH=" ; NNA
80 INPUT "WEIGHT OF 10-ML H2SO4=" ; W0
90 INPUT "QANTITY OF NAOH V0=" ; V0
100 N0 = NNA * V0 / 1000 : M = 98
110 WWH = W0 - N0 * M / 2
120 G0 = N0 / WWH
130 INPUT "WEIGHT OF SOLUTION IN THE CATHODE=" ; WS
140 INPUT "WEIGHT OF 10-ML SOLUTION IN THE
```

```

      CATHODE=" , W1
150 INPUT "QUANTITY OF NAOH V1=" , V1
170 N1=NNA * V1/1000 : WH= W1- N1 * M/2
180 G1=N1/WH : WA= WS * WH/W1
190 TRAN = (G0- G1) * WA/Q
200 PRINT "TRAN. OF SO4=" , TRAN
210 PRINT "TRAN. OFH=" , 1- TRAN
220 END

```

例 由下列实验数据计算 SO_4^{2-} 与 H^+ 的迁移数。室温 $19(^{\circ}\text{C})$ ，大气压 $762.2(\text{mmHg})$ 。水的饱和蒸气压 $P_w = 16.5(\text{mmHg})$ ，混合气体体积 $V = 17.52(\text{ml})$ ， NaOH 溶液浓度 $0.05233(\text{mol/L})$ ，通电前 $10\text{ml H}_2\text{SO}_4$ 溶液重 $10.1(\text{g})$ ，滴定用 NaOH 溶液 $V_0 = 6.74(\text{ml})$ ，通电后阴极区溶液总重量 $56.0(\text{g})$ ，通电后 10ml 阴极区 H_2SO_4 溶液重 $10.0(\text{g})$ ，滴定该溶液用 NaOH $6.05(\text{ml})$ 。

运行结果

```

T = 19.0
P = 762.2
VAPOR PRESSURE OF WATER = 16.5
V = 17.52
FARADAYS OF Q = 9.56025641E - 04
CONCENTRATION OF NAOH = .05233
WEIGHT OF 10 - ML H2SO4 = 10.1
QUANTITY OF NAOH V0 = 6.74
WEIGHT OF SOLUTION IN THE CATHODE = 56.0
WEIGHT OF 10 - ML SOLUTION IN THE CATHODE = 10.0
QUANTITY OF NAOH V1 = 6.05
TRAN OF SO4 = .191375947
TRAN OF H = .808624053

```

§ 1.3 凝固点降低法测分子量

根据稀溶液的依数性，溶剂凝固点降低与溶质分子量之间有如下关系：

$$M = K_f \cdot \frac{1000 \cdot W_{\text{质}}}{\Delta T_f \cdot W_{\text{剂}}} \quad (10)$$

其中， K_f 为凝固点降低常数； $W_{\text{质}}$ 、 $W_{\text{剂}}$ 分别代表溶质和溶剂的重量； ΔT_f 为凝固点下降值。

例如，以苯为溶剂，测萘的分子量。已知苯的 $K_f = 5.10$ 。用贝克曼温度计测温差（要求所得分子量的最大相对误差小于30%）。

记录的初始数据有：

室温 $t(^{\circ}\text{C})$ 。

萘的重量 $W(\text{g})$ （称量误差约为0.0002g）。

苯的用量 $V(\text{ml})$ （用移液管移取，相对误差约为0.3%）。

苯的重量 WB 由下式计算：

$$WB = (d_0 - 1.0636 \times 10^{-3}t)V \quad (11)$$

其中， d_0 是 0°C 时苯的密度，其值为 $0.9001(\text{g}/\text{cm}^3)$ 。 $t^{\circ}\text{C}$ 时苯的密度由下式计算：

$$d_t = d_0 - 1.0636 \times 10^{-3}t \quad (12)$$

设对纯溶剂和溶液各测量 N 次，溶剂的凝固点读数为 $T_1, T_2, T_3, \dots, T_N$ 。平均值用 T 表示；溶液的凝固点读数为 $TL_1, TL_2, \dots, TL_N$ ，平均值用 TL 表示。凝固点降低值 ΔT 由下式计算：

$$\Delta T = T - TL$$

校正 ΔT 得出正确的凝固点降低值 ΔT_f ：

$$\Delta T_f = \Delta T \cdot f + 1.6 \times 10^{-4} \Delta T (t' + T + TL - t) \quad (13)$$

式中, t' 是贝克曼温度计的调整温度, f 值需查表。

例 当 $t' = 5^\circ\text{C}$ 时 $f = 0.995$ (见物理化学实验教材有关贝克曼温度计的使用方法)。

分子量的相对误差

$$\frac{\Delta M}{M} = \frac{\Delta V}{V_{\text{苯}}} + \frac{\Delta W}{W_{\text{苯}}} + \frac{\Delta(\Delta T)}{T} \quad (14)$$

已知 $\Delta V/V \approx 0.3\%$, $\Delta W \approx 0.0002$

$$\Delta(\Delta T) = \left| \frac{\sum_{i=1}^N |T_i - T|}{N} + \frac{\sum_{i=1}^N |TL_i - TL|}{N} \right|$$

萘的分子量 = $M \pm \Delta M$

程序名: MWFP

程序中各变量的意义:

数组 $T(N)$ 存放溶剂的凝固点读数 T_i 。

数组 $TL(N)$ 存放溶剂的凝固点读数 TL_i 。

N ——溶剂和溶液的测量次数。

TR ——表示室温。

DD ——表示 $\Delta(\Delta T)$ 。

DT ——表示 ΔT_f 。

F ——(13)式中的 f 。

TB ——(13)式中的 t' 。

KF ——表示 K_f 。

T_1 、 T_2 分别表示 $\sum_{i=1}^N |T_i - T|$ 和 $\sum_{i=1}^N |TL_i - TL|$ 。

其余各量程序中已有说明。

程序清单

```
10 INPUT "SET OF DATA N=" ; N
15 DIM T(N), TL(N)
20 INPUT "ROOM TEMPERATURE TR=" ; TR
30 INPUT "TEMPERATURE FOR SETTING THE
   BECKMANN THERMOMETER TB=" ; TB
40 INPUT "F=" ; F
50 INPUT "THE WEIGHT OF NAPHTHALENE W=" ; W
60 INPUT "AMOUNT OF BENZENE V=" ; V
70 KF=5.10 : D0=0.9001
75 T=0 : TL=0 : T1=0 : T2=0
80 FOR I=1 TO N
90 READ T(I), TL(I)
100 T=T+T(I) : TL=TL+TL(I)
110 NEXT I
120 T=T/N : TL=TL/N
130 FOR I=1 TO N
140 T1=T1+ABS(T(I)-T) : T2=T2+ABS(TL(I)-TL)
150 NEXT I
160 DD=(T1+T2)/N : DT=ABS(TL-T)
170 DT=DT*F+1.6E-4*DT*(TB+T+TL-TR)
200 WB=(D0-1.063E-3*TR)*V
210 M=KF*1000*W/(DP*WB)
220 EM=3E-3/V+2E-4/W+DD/DT
230 IF EM>0.03 THEN PRINT "YOUR RESULT IS
   BAD, REDO PLEASE!"
240 PRINT "MOLECULAR WEIGHT OF NAPHTHALENE
   M=" ; M
250 PRINT "RELATIVE ERROR EM=" ; EM
```